VA for a sech soliton using a Gaussian ansatz

```
> restart;
> interface(showassumed=0):;
```

Ansatz of the form u(x,t) = A(x,t) * exp(I*phi(x,t))

```
> A := B(t)*exp(-(x)^2/(2*w(t)^2));
> phi := -mu*t;
```

$$A := B(t)\, e^{-\frac{x^2}{2\, w(t)^2}}$$

$$\phi := -\mu\, t \tag{1}$$

Potential

```
> V := 0;
```

$$V := 0 \tag{2}$$

Lagrangian: (I/2)(u u*_t - u_t u*) + (1/2) u_x u*_x - (1/2) |u|^4 = A^2 phi_t + (1/2)* (A_x^2+A^2 phi_x^2) -(1/2)*A^4+V*A^2

```
> LA := A^2*diff(phi,t) + (1/2)*(diff(A,x)^2+A^2*diff(phi,x)^2) -
  (1/2)*A^4 + V*A^2;
```

$$LA := -B(t)^2 \left(e^{-\frac{x^2}{2\, w(t)^2}}\right)^2 \mu + \frac{B(t)^2\, x^2 \left(e^{-\frac{x^2}{2\, w(t)^2}}\right)^2}{2\, w(t)^4} - \frac{B(t)^4 \left(e^{-\frac{x^2}{2\, w(t)^2}}\right)^4}{2} \tag{3}$$

Take out t-dependence so we can compute Leff

```
> sub1 := diff(B(t),t)=Bp,diff(w(t),t)=wp:
> sub2 := B(t)=B,w(t)=w:
> LA2 := subs({sub1,sub2},LA);
```

$$LA2 := -B^2 \left(e^{-\frac{x^2}{2\, w^2}}\right)^2 \mu + \frac{B^2\, x^2 \left(e^{-\frac{x^2}{2\, w^2}}\right)^2}{2\, w^4} - \frac{B^4 \left(e^{-\frac{x^2}{2\, w^2}}\right)^4}{2} \tag{4}$$

Leff = integral of Lag, we need to assume that a>0 and xi real to be able to evaluate integrals

```
> assume(w>0);
> Leff := expand(int(LA2,x=-infinity..infinity));
  LeffR := expand(Leff*4/sqrt(Pi));
```

$$Leff := -B^2\, \mu\, w \sqrt{\pi} + \frac{B^2 \sqrt{\pi}}{4\, w} - \frac{B^4 \sqrt{2}\, w \sqrt{\pi}}{4}$$

$$LeffR := -4\, B^2\, \mu\, w + \frac{B^2}{w} - B^4 \sqrt{2}\, w \tag{5}$$

Put back t-dependences so that we can do Euler-Lag

```
> subi1 := Bp=diff(B(t),t),wp=(diff(w(t),t)):
> subi2 := B=B(t),w=w(t):
```

Euler-Lagrange eqs

```
> dLwp := subs({subi1,subi2},diff(LeffR,wp)):;
> dLBp := subs({subi1,subi2},diff(LeffR,Bp)):;
> eq1 := diff(dLwp,t)=subs({subi1,subi2},diff(LeffR,w));
> eq2 := diff(dLBp,t)=subs({subi1,subi2},diff(LeffR,B));
```

$$eq1 := 0 = -4\,B(t)^2\,\mu - \frac{B(t)^2}{w(t)^2} - B(t)^4\sqrt{2}$$

$$eq2 := 0 = -8\,B(t)\,\mu\,w(t) + \frac{2\,B(t)}{w(t)} - 4\,B(t)^3\sqrt{2}\,w(t) \qquad \textbf{(6)}$$

Sove Euler-Lag ODEs simultaneously

```
> eq3:=simplify(eq1*w(t)^2/B(t)^2-eq2*w(t)/(4*B(t)));
  solmu:=solve(eq3,mu);
  solw:=solve(subs({mu=solmu},eq1),w(t));
```

$$eq3 := 0 = -2\,w(t)^2\,\mu - \frac{3}{2}$$

$$solmu := -\frac{3}{4\,w(t)^2}$$

$$solw := \frac{2^{1/4}}{B(t)}, -\frac{2^{1/4}}{B(t)} \qquad \textbf{(7)}$$

```
>
```